Übersicht Kosten- und Preistheorie

Dokumentnummer: D1905

Inhaltsverzeichnis

Begriffe

In der Kosten- und Preistheorie geht es um
a) Kosten (zusammengesetzt aus variablen Kosten und Fixkosten)
b) Preis
c) Erlös oder Umsatz (es gibt keinen fixen Erlös oder Umsatz)
d) Gewinn

Kostenfunktionen

Typische Kostenfunktionen sind
a) die llnerare Kostenfunktion K = k.x + F,
b) die quadratische Kostenfunktion K = a.x^2 + b.x + c und
c) die kubische Kostenfunktion K = a.x^3 + b.x^2 + c.x +d (s-förmiger Kostenverlauf).

Beispiele für lineare, quadratische und kubische Kostenfunktion

(%i10) LK(x):=3*x+36;

(%o10) $\mathrm{LK}(x) := 3\,x + 36$

(%i11) QK(x):=x**2+8*x+36;

(%o11) $\mathrm{QK}(x) := x^2 + 8\,x + 36$

(%i12) KK(x):=0.1*x**3-2*x**2+16*x+36;

(%o12) $\mathrm{KK}(x) := 0.1\,x^3 - 2\,x^2 + 16\,x + 36$

Kostenauflösung für lineare, quadratische und kubische Kostenfunktion
(= Aufteilung in fixe und variable Kosten

(%i13) AufteilungLK:[LK(0),LK(x)-LK(0)];

(%o13) $[36, 3\,x]$

(%i14) AufteilungQK:[QK(0),QK(x)-QK(0)];

(%o14) $[36, x^2 + 8\,x]$

(%i15) AufteilungKK:[KK(0),KK(x)-KK(0)];

(%o15) $[36, 0.1\,x^3 - 2\,x^2 + 16\,x]$

Grafik lineare, quadratische und kubische Kostenfunktion

(%i16) wxplot2d([LK(x),QK(x),KK(x)], [x,0,15])$

(%t16)



Grafik lineare Kostenfunktion

(%i17) wxplot2d([LK(x)], [x,0,15])$

(%t17)



Grafik quadratische Kostenfunktion

(%i18) wxplot2d([QK(x)], [x,0,15])$

(%t18)



Grafik kubische Kostenfunktion

(%i19) wxplot2d([KK(x)], [x,0,15])$

(%t19)



Bestimmung der langfristigen Preisuntergrenze.
Die langfristige Preisuntergrenze ist das Minimum der Durchschnittskosten.
Die Menge, bei der die Durchschnittskosten ein Minimum annehmen, heißt Betriebsoptimum.

(%i20) D:LK(x)/x /* Sonderfall lineare Kostenfunktion */;

(%o20) $\frac{3\,x+36}{x}$

(%i21) BO:limit(D,x,inf);

(%o21) 3

(%i22) LP:D,x=BO;

(%o22) 15

(%i23) D:QK(x)/x /* Beispiel quadratische Kostenfunktion */;

(%o23) $$\frac{x^2 + 8x + 36}{x}$$

(%i24) ab:diff(D,x);

(%o24) $$\frac{2x + 8}{x} - \frac{x^2 + 8x + 36}{x^2}$$

(%i25) l:realroots(ab=0);

(%o25) $[x = -6, x = 6]$

(%i26) BO:x,l[2];

(%o26) 6

(%i27) LP:D,x=BO;

(%o27) 20

(%i28) D:KK(x)/x /* Beispiel kubische Kostenfunktion */;

(%o28) $$\frac{0.1x^3 - 2x^2 + 16x + 36}{x}$$

(%i29) ab:diff(D,x);

(%o29) $$\frac{0.3x^2 - 4x + 16}{x} - \frac{0.1x^3 - 2x^2 + 16x + 36}{x^2}$$

(%i30) l:realroots(ab=0);

(%o30) $$\left[x = \frac{382116847}{33554432}\right]$$

(%i31) BO:x,l,numer;BO:floor(BO*100+0.5)/100.0;

(%o31) 11.38796946406364

(%o32) 11.39

(%i33) LP:D,x=BO;LP:floor(LP*100+0.5)/100.0;

(%o33) 9.353877251975417

(%o34) 9.35

Bestimmung der kurzfristigen Preisuntergrenze.
Das Betriebsminimum ist jene Produktionsmenge, bei der die durchschnittlichen variablen
Kosten am kleinsten werden.
Die kurzfristige Preisuntergrenze ist das Minimum der durchschnittlichen variablen Kosten.
Zur Ermittlung muss man das Betriebsminimum in die durchschnittlichen variablen Kosten
einsetzen.

(%i35) VKK(x):=KK(x)-KK(0) /* die Fragestellung ist nur bei kubischer Kostenfunktion sinnvoll */;

(%o35) $\mathrm{VKK}(x) := \mathrm{KK}(x) - \mathrm{KK}(0)$

(%i36) D:VKK(x)/x;

(%o36) $\dfrac{0.1\,x^3 - 2\,x^2 + 16\,x}{x}$

(%i37) ab:diff(D,x);

(%o37) $\dfrac{0.3\,x^2 - 4\,x + 16}{x} - \dfrac{0.1\,x^3 - 2\,x^2 + 16\,x}{x^2}$

(%i38) l:realroots(ab=0);

(%o38) $[\,x = 10\,]$

(%i39) BM:x,l /* das ist das Betriebsminimum */;

(%o39) 10

(%i40) KP:D,x=BM /*das ist die kurzfristige Preisuntergrenze */;

(%o40) 6.0

Preis

Beim Preis gibt es 3 typische Fälle:
a) p = const,
b) die lineare Nachfragefunktion p = a.x + b und
c) die quadratische Nachfragefunktion $p = a.x^2 + b.x + c$.

Bei der linearen Nachfragefunktion muss man oft
a) die Preisobergrenze,
b) die Sättigungsmenge bestimmen.

In der Preisobergrenze sinkt die Nachfrage auf x = 0.
Die Sättigungsmenge ist die Nachfrage, wenn der Preis p = 0 ist.

Ermittlung einer linearen Nachfragefunktion
Beispiel:
Bei einem Lospreis von 1 € können 2000 Lose abgesetzt werden,
bei einem Lospreis von 2 € sinkt die Nachfrage auf 1400 Lose.
Welche lineare Nachfragefunktion passt zu diesen Daten?

(%i41) p1:1;x1:2000;p2:2;x2:1400;

(%o41) 1

(%o42) 2000

(%o43) 2

(%o44) 1400

(%i45) g(x,p):=p=a*x+b;

(%o45) $g(x, p) := p = a\, x + b$

(%i46) g1:g(x1,p1);

(%o46) $1 = b + 2000\, a$

(%i47) g2:g(x2,p2);

(%o47) $2 = b + 1400\, a$

(%i48) l:solve([g1,g2],[a,b]);

(%o48) $[\,[\, a = -\frac{1}{600}, b = \frac{13}{3}\,]\,]$

(%i49) Nachfragefunktion:p=a*x+b,l[1];

(%o49) $p = \frac{13}{3} - \frac{x}{600}$

Resultat zur Funktion verarbeiten

(%i50) term:rhs(Nachfragefunktion);

(%o50) $\frac{13}{3} - \frac{x}{600}$

(%i51) p(x):=''term;

(%o51) $p(x) := \frac{13}{3} - \frac{x}{600}$

Bestimmung der Randpunkte

(%i52) po:p(0) /* die Preisobergrenze */;

(%o52) $\frac{13}{3}$

(%i53) l:realroots(p(x));

(%o53) $[x = 2600]$

(%i54) xs:x,l /* die Sättigungsmenge */;

(%o54) 2600

Grafik dieser Nachfragefunktion

(%i55) wxplot2d([p(x)], [x,0,xs], [y,0,po])$

(%t55)



Erlös oder Umsatz

Der Erlös oder Umsatz ist das Produkt aus Menge mal Preis.
Daher ergeben sich 3 Fälle:
a) die lineare Umsatzfunktion U = p.x,
b) die quadratische Umsatzfunktion U = $a.x^2 + b.x$ und
c) die kubische Umsatzfunktion U = $a.x^3 + b.x^2 + c.x$.

Bei einer linearen Nachfragefunktion erhalten wir eine quadratische Erlösfunktion (Umsatzfunktion).

(%i56) p(x);

(%o56) $\frac{13}{3}-\frac{x}{600}$

(%i57) U(x):=p(x)*x;

(%o57) $U(x):=p(x)\,x$

(%i58) U(x),expand;

(%o58) $\frac{13\,x}{3}-\frac{x^2}{600}$

Grafik der Umsatzfunktion

(%i59) wxplot2d([U(x)], [x,0,xs])$

(%t59)



Ermittlung des maximalen Umsatzes

Ermittlung des maximalen Umsatzes
Wenn eine quadratische (das ist der häufigste Fall) oder eine kubische Umsatzfunktion vorliegt, sind die folgenden Fragestellungen interessant:

a) umsatzmaximaler Preis,
b) umsatzmaximale Menge und
c) maximaler Umsatz.

Für die Bestimmung der umsatzmaximalen Menge muss man die erste Ableitung der Umsatzfunktion NULL setzen.
Den umsatzmaximalen Preis erhält man durch Einsetzen in die Nachfragefunktion und den maximalen Umsatz durch Einsetzen in die Umsatzfunktion (Erlösfunktion).

(%i60) p(x);U(x) /* Nachfragefunktion und Umsatzfunktion */;

(%o60) $\frac{13}{3}-\frac{x}{600}$

(%o61) $\left(\frac{13}{3}-\frac{x}{600}\right)x$

(%i62) ab:diff(U(x),x);

(%o62) $\frac{13}{3}-\frac{x}{300}$

(%i63) l:realroots(ab);

(%o63) $[\, x = 1300 \,]$

(%i64) xu:x,l /* umsatzmaximale Menge */;

(%o64) 1300

(%i65) p(xu),numer; floor(%*100+0.5)/100.0 /* umsatzmaximaler Preis */;

(%o65) 2.166666666666666

(%o66) 2.17

(%i67) U(xu),numer;floor(%*100+0.5)/100.0 /*maximaler Umsatz */;

(%o67) 2816.666666666666

(%o68) 2816.67

Gewinn

Der Gewinn ist die Differenz aus Erlös minus Kosten.
Die Gewinnfunktion in unseren Aufgaben ist meistens
a) kubisch G = $a.x^3 + b.x^2 + c.x + d$ könnte aber auch
b) quadratisch G = $a.x^2 + b.x + c$ sein. Praktisch gar nicht vorkommt
c) eine lineare Gewinnfunktion.

Fragestellungen sind:
a) gewinnmaximale Menge,
b) maximaler Gewinn,
c) gewinnmaximaler Preis und außerdem
d) Nutzenschwelle und
e) Nutzengrenze.

```
(%i69) U(x):=5*BO*x;
(%o69) U(x):= 5 BO x

(%i70) U(x);QK(x);
(%o70) 56.95 x
(%o71) x^2 + 8 x + 36
```

Erstellung der Gewinnfunktion

```
(%i72) G(x):=U(x)-QK(x),expand;G(x);
(%o72) G(x):= U(x) - QK(x)
(%o73) - x^2 + 48.95 x - 36
```

Ermittlung des maximalen Gewinns

```
(%i74) ab:diff(G(x),x);
(%o74) 48.95 - 2 x

(%i75) l:realroots(ab),numer;
(%o75) [ x = 24.47499999403954 ]

(%i76) xC:x,l;xC:floor(xC*100+0.5)/100.0 /* die gewinnmaximale Menge */;
(%o76) 24.47499999403954
(%o77) 24.47
```

(%i78) Gmax:G(xC);Gmax:floor(Gmax*100+0.5)/100.0 /* der maximale Gewinn */;

(%o78) 563.0255999999999

(%o79) 563.03

Grafische Darstellung von Umsatz, Kosten und Gewinn

(%i80) wxplot2d([U(x),QK(x),G(x)], [x,0,2*xC])$

(%t80)



Deckungsbeitrag

Der Deckungsbeitrag ist entweder die Summe aus Gewinn plus Fixkosten oder die Differenz aus Umsatz minus variable Kosten. Damit ist auch die Deckungsbeitragsfunktion häufig kubisch (vergl. mit den Ausführungen Gewinnfunktion).

(%i81) kill(all);

(%o0) *done*

(%i1) K(x):=V(x)+F;

(%o1) $K(x):=V(x)+F$

(%i2) G(x):=U(x)-K(x);

(%o2) $G(x):=U(x)-K(x)$

(%i3) DB(x):=G(x)+F;DB(x);

(%o3) $DB(x):=G(x)+F$

(%o4) $U(x)-V(x)$

(%i5) DB(x):=U(x)-V(x);DB(x);

(%o5) $DB(x):=U(x)-V(x)$

(%o6) $U(x)-V(x)$

Created with wxMaxima.